

Általános Földtani Szemle 24, 211-222, Budapest, 1989

GEOTERMIKUS ADATOK FELDOLGOZÁSA COMMODORE-64 SZÁMÍTÓGÉPPCEL

Processing of geothermal data by Commodore 64 computer

GYÓRY Dóra

A szerző rövid történeti áttekintés után Dombóvár környékének geotermikus viszonyait vizsgálva mutatja be az általa írt program felhasználási lehetőségét. Részletesen ismerteti a geotermikus gradiens számításának módszereit, ezek előnyét és hátrányát is jelezve. A program a talphőmérsékleteket mélység függvényében ábrázolja, a pontokra görbét illeszt; fúrásonként meghatározza a geotermikus gradiens értékét és ezek eloszlását hasábdiaagram segítségével szemlélteti. A program használhatóságát növeli a függelékben közölt folyamatábra és magyarázat.

The computer program written by the author helps in geothermal gradient calculations. Several methods are compared and discussed.


The programme plots bottom-hole temperatures as function of depth and fits a regression curve. Then geothermal gradient values are determined for each borehole and plotted in a frequency diagram. The flow chart and list of the programme with explanations are given as supplements.

A geotermikus gradiens

A felszíntől a Föld belseje felé haladva általában 20 m mélységig csökken a hőmérséklet, utána növekszik (SCHEFFER V.- KÁNTÁS K. 1949). A növekedés mértékét a geotermikus gradiens fejezi ki, amely az 1 km-ről eső hőmérséklet-növekedést adja meg; mértékegysége $^{\circ}\text{C}/\text{km}$. Használják még a geotermikus gradiens reciprokát, ill. a geotermikus mélységlépcső kifejezést is. Ebben az esetben az 1 $^{\circ}\text{C}$ hőmérséklet-változáshoz szükséges mélységintervallumot kapjuk meg. /A szakirodalomban egészen a 70-es évek elejéig az utóbbi volt az uralkodó./

Történeti áttekintés

A kevés és többnyire megbízhatatlan adat ellenére PAPP K. (1919) már a század elején megállapította, hogy "meglepő a Nagy Magyar Alföld általános tetemes melegsége, amelynek eredménye a csekély gradiens" /10-22 $\text{m}/^{\circ}\text{C}$ /. Megállapítása a későbbi kutatások révén igazolódott. SÜMEGHY J. (1929), BÉLTEKI L. (1963, 1965, 1966) munkássága szolgáltatott nagy mennyiségű hőmérséklet adatot az egész ország területére. BÉLTEKY L. kezdeményezésére 1956 óta a 300 m-nél; 1963 óta minden 200 m-nél mélyebb vízkutató fúrásban kötelező mérni a



1. ábra. A vizsgált terület

Fig. 1. The Dombóvár region in Hungary

talphőmérsékletet. 1965-től a VITUKI által szerkesztett "Magyarország hévízkútjai" c. kiadvány biztosítja a folyamatos adatszolgáltatást és -feldolgozást.

Számos kiértékelő munka született, amelyek közös, még érvényes megállapítása, hogy a preneogén aljzat domborzata és a geotermikus gradiens között kimutatható a kapcsolat, ill. a felszín alatti áramlások befolyásolják a hőteret (DÜVÉNYI P. et al. 1982). A magyarországi átlagos reciprokl geotermikus gradiens $18 \text{ m}^\circ\text{C}$ (BÉLTEKY L. 1965, 1966, BOLDIZSÁR T. 1964), ami a világtátlagnak számító $30 \text{ m}^\circ\text{C}$ jóval kisebb.

SCHEFFER V.- KÁNTÁS K. (1949) a Dunántúl regionális geofizikájának vizsgálatakor megállapították, hogy ott,

ahol a paleozoós kőzetek a felszínhez közelebb vannak, anomálishan kicsik a mélységlépcső értékek, míg a mezozoós kőzetek felett ellentétes anomália mutatható ki. Így a kis gradiens zónája a paleozoikum képét követi. Ez következik abból is, hogy a paleozoós kőzetek lényegesen jobb hővezető-képességűek, mint az üledékes kőzetek. KORIM K. (1973) szerint a Dunántúli-középhegységtől D-re pozitív geotermikus anomáliájú területek vannak. SÜMEGHY J. (1929) a Kapos vonalával közel párhuzamos ternális anomália vonalat mutatott ki. NÉMEDI-VARGA Z. (1977) a Kapos-völgyében a Dunántúl geotermikus átlagától ($20\text{--}24 \text{ m}^\circ\text{C}$, SCHMIDT E.R. et al. in NÉMEDI-VARGA Z. 1977) eltérő értékeket ($13,2\text{--}17,7 \text{ m}^\circ\text{C}$) számított, ami alátámasztja a Kapos-vonal létezését.

A geotermikus gradiens sz nol sakor felmer l  probl n k

A f r si tevek nysege / raml  iszap/ megv ltoztatja a lyuk k rnyezet nek h m rs klet t. A r teg t nylegesen h m rs klet t akkor kapn nk meg, ha a f r s befejez se ut n lez rt k tban, megfelel  id  eltelt vel t rt nne a m r s. /Erre az id tartamra k l nb z  szarv k k l nb z  becsl seket adnak; n h ny napt l 2-3  vet, a f ldtani  s f r stechnol giai viszonyokt l f gg en (CSEREMENSKIJ in D V NYI P. et al. 1983). A v zkutat  f r sokn l  ltal ban nincs lehet s g a hosszadalmas v rakozezsra, a k l nb z  technikai m veletek elv gz se ut n t rt nik a m r s. Így a legt bb talph m rs klet-adat kisebb h m rs kletet mutat, mint a val di  rt k, a bel le sz m tott val s gos geotermikus gradiens  rt ke sem t kr zi a t nyleges  rt ket, val sz n leg kisebbek ann l. Nehez ti az adatfeldolgoz st, hogy ezeket jelent s m r si hiba is terhelheti. A hibalehet s gek k z  tartozik:

1, a h m r  saját  s olvas si hib ja
2, ha a h m r  nincs elegend  ideig a vizsg lt m lys gben, nem tudja  tvenni a k rnyezet h m rs klet t

3, a h m r  a falh z s során ler z dhat
4, a nyom s n veked s nek hatás ra a higanysz l elmozdulhat.

(D V NYI P. et al. 1983). G LFI J.  s LIEBE P. (1977) a maximum-h m r vel t rt n  m r sesek hib j t $\pm 2,5$  C-ra becs lik, de hozz teszik, hogy kedvez tlen esetben enn l j val nagyobb is lehet.

A m r s hibalehet s ge l nyegesen kisebb, ha a felsz nen t rt nik a v z h m rs klet nek meghat roz sa. Ebben az esetben viszont figyelembe kell venni, hogy a k tban felfel  haladva a v z t bb  - keveseb  leh l. T bb m dszert is kidolgoztak az egyens lyi h m rs kletet j l k zel t   rt k meghat roz s ra (BOLDIZS R T. 1960, SAL T P. 1964, G LFI J.- LIEBE P. 1977). Ezek k z l csak a Liebe-f le  s szef gg st ellen rizt k

nagy sz m  m lys gi h m rs klet-m r ssel. /Sz n t sainban  n is ezt haszn ltam - l  mell klet/


A kifoly  v z h m rs klet t szinte minden k tn l megm rik, csak egy-k t, a sz zad elej n f rt k t esetén hi nyzik ez az adat. Korrekci t viszont csak azokn l a f r sokn l  rdeemes v gezni, ahol a kifoly  v z  s a talajv z h m rs klete k z t elegend  nagy a k l nb s g.  ltal ban a 250 m-n l nagyobb m lys gb l sz rmarz  vizek h m rs klet adat t korrig lj k (G LFI J. - STEGENA L. 1977, D V NYI P. et al. 1983), de mivel a felsz ni k z ph m rs klet  tlagosan 10-11  C, ez rt a 20  C ill. enn l magasabb v zh m rs kletek esetén m r elegend  a k l nb s g ahhoz, hogy  rdeemes legyen korrekci t v gezni (LIEBE P. sz beli k zl se).

Domb v r k rnyek n m ly tett v zkutat  f r sok adatainak feldolgoz sa

A vizsg lt területen /l.  bra/ 175 f r s m ly lt. Az adatok feldolgoz s t Commodore 64 sz m t g p seg ts g vel v geztem el ( RY L. 1985). A programban (l. mell klet) az  br z l shoz a Simon's Basic seg gprogramot (PLENGE - SZCZEPANOVSKI 1983) haszn ltam fel. A 175 f r s k z l csak 62-n l tal ltam talph m rs klet adatot. Ezeket a m lys g f ggv nyben  br zoltam  s a pontokra g rb t illesztettem. A g rbe elegend  j  illeszked s t mutatja, hogy a regresszi   rt ke 0,89. /2.  bra/ Minden egyes f r s esetén kisz moltam a g rb nek megfelel  h m rs klet  rt ket is.

A talph m rs letb l sz m tottam ki a val s gos geotermikus gradienst. Nagys grendi ellen rz s re haszn ltam fel a kifoly  v z h m rs klet b l korrekci  ut n meghat rozott l tsz lagos geotermikus gradienst /l. mell klet/. Ahol a k t  rt k k z t t l nagy volt a k l nb s g /11  C/km-n l nagyobb/, ott val sz n leg m r si hiba lehets ges,  gy azokat a szerkeszt s n l nem vettem figyelembe. A gradiensek eloszl s t

Y=A+B1*X+B2*X^{10.5} EGYENLET PARAMETEREI:
 B1= .0162579411
 B2= 1.35253757
 A= 2.48779428
 R2= .891771356



2. ábra. A talphőmérsékletek a mélység függvényében és a regressziós görbe.
 Fig. 2. Bottom-hole temperatures vs. depth and the regression curve

vizsgálva, mindkét esetben Gauss-görbét kaptam. /3-4. ábra/ A valóságos geotermikus gradiens átlaga $66,6 \text{ } ^\circ\text{C}/\text{km}$, ami $15 \text{ m}/^\circ\text{C}$ reciprok értéknek felel meg. A látszólagos geotermikus gradiens átlaga egy kicsit nagyobb, $68,2 \text{ } ^\circ\text{C}/\text{km}$.


Térképi ábrázolásnál csak azokat az értékeket vettem figyelembe, ahol a valóságos és látszólagos geotermikus gradiens között az eltérés $10 \text{ } ^\circ\text{C}/\text{km}$ -nél kisebb. Ebben az esetben a talphőmérsékletből számított értéket ábrázoltam. Az adatszelekció eredményeképpen alig maradt olyan adat, amit az izovonalas térkép szerkesztéséhez felhasználhattam, és ezek eloszlása is egyenlőtlen. Ezért az "adathiányos" területeken kiegészítésként számításba vettem azokat a fúrásokat is, ahol csak a valóságos geotermikus gradiens határozható meg.

Az átlagnál $/60 - 70 \text{ } ^\circ\text{C}/\text{km}/$ kisebb


értékkel jellemezhető területeken /a Kapostól északra Gölle és Nak környezetében, ill. a Kapostól délre/ a felszínről mélybe szivárgó víz hatása érvényesül. Ezzel szemben Döbrököz - Dalmand környékén erőteljes feláramlás mutatható ki $/70 \text{ } ^\circ\text{C}/\text{km}$ - nél jóval magasabb GG(v) értékek dominálnak/. 5. ábra / A térképen behúzott izovonalak az adatok kis száma és egyenlőtlen eloszlása miatt csak feltételezettek./

Összefoglalás


Egy terület geotermikus viszonyainak tisztázása hozzásegíthet a mélységi vizek áramlási irányainak felderítéséhez. Dombóvár környékén elkülöníthető a beszivárgási és a feláramlási terület, de az áramlási rendszerek pontosításához szükséges lenne nagyobb terület átfogó kutatása, egyéb vizsgálatokkal kiegészítve.



3. ábra. A valószínűségi geotermikus gradiens eloszlása
 Fig. 3. Frequency of true geothermal gradients.



4. ábra. A látszólagos geotermikus gradiens eloszlása.
 Fig. 4. Frequency of virtual geothermal gradients.



5. ábra. A vizsgált terület geotermikus viszonyai
 Fig. 5. Map of geothermal gradients around Dombóvár

- BÉLTEKY L. (1963): Magyarország területének geotermikus viszonyai a legújabb vízfeltáró fúrások alapján Hidrológiai Közlöny 1963, 5.sz. pp.401-411
- BÉLTEKY L. (1963): Újabb adatok a hazai geotermikus vizsgálatokhoz Geofizikai Közlemények 1963, 12k. 1-2.sz. pp.3-47
- BÉLTEKY L. (1965): Magyarország geotermikus viszonyai /Magyarország hévízkútjai/ VITUKI, 1965 Budapest, pp.59-67
- BÉLTEKY L.(1966): Magyarország geotermikus viszonyai az 1956-65 évi kútfúrásokban végzett hőmérsékletmérések alapján Hidrológiai Közlöny 1966, 9.sz. pp.406-412
- BÉLTEKY L.- KORIM K.- LIEBE P. (1971): Magyarország hévízkútjai /Hévízkút-kataszter II./ VITUKI, 1971 Budapest, pp.1-252
- BOLDIZSÁR T. (1960): Geotermikus vizsgálatok a Nagy Magyar Alföldön Bányászati Lapok 1960, 93.k. 5.sz, pp.306-309
- BOLDIZSÁR T.(1964): Magyarország geotermikus térképe és földi hőárama MTA Műszaki Tudományos Osztályának Közleményei 1964, 33.k, pp.307-327
- DÖVÉNYI P.- HORVÁTH F.- LIEBE P.- GÁLFI J.- ERKI I. (1903): Magyarország geotermikus viszonyai Geofizikai Közlemények 1983, 29.k, 1.sz. pp.1-114
- GÁLFI J. és LIEBE P. (1977): Magyarország geotermikus hőmérséklet-térképei in STEGENA L.: Geotermikus rezervoárkutató I. Jelentés a KFH részére, ELTE TTK, 1977 Budapest, pp.1-93
- GÁLFI J.- LIEBE P. (1977): Magyarország geotermikus hőmérséklet-térképei a vízfeltáró fúrások alapján Kutatási Jelentés, VITUKI 1977
- GÁLFI J. - STEGENA L. (1977): Magyarország geotermikus viszonyai /Magyarország hévízkút-katasztere III./ VITUKI, 1977 Budapest, pp.44-47
- KORIM K.(1973): Magyarország geotermikus viszonyai Mérnökgeológiai Szemle, 1973.jan. 12.k. pp.27-40
- NÉMEDI VARGA Z.(1977): A Kapos-vonal Földtani Közlöny, 1977, 107.k. pp.313-328
- LORBERER Á. (1985): Nagydorog környékének hévízföldtani viszonyai VITUKI, 1985 Budapest, kézirat
- PAPP K. (1919): A földmelegségi mélységi fokozat (geotermikus gradiens) Magyarországon in. SCHAFFER X. F.: Általános Geológia, 1919 Budapest, pp.623-628
- PLENGE - SZCZEPANOVSKI (1983): Das Trainingbuch zum Simon's Basic DATA Becker, Düsseldorf, 1983
- SALÁT P. (1964): Az artézi kutak vizének lehűlése kifolyáskor Magyar Geofizika 1964, 5.k. 1-2.sz. pp.36-96
- SCHEFFER V. - KÁNTÁS K.(1949): A Dunántúl regionális geofizikája Földtani Közlöny 1949, 79.k. 9-12.sz. pp.327-356
- SÜMEGHY J. (1929): Die Geothermischen Gradienten des Alföld Magyar Királyi Földtani Intézet Évkönyve 1929, 28k. 3.sz. pp.273-370
- ÚRY L.(1985): Commodore 64 /BASIC felhasználói kézikönyv/ LSI Alkalmazástechnikai Tanácsadó Szolgálat, Budapest, 1985

```

10 PRINT "□"
20 PRINT " *****"
30 PRINT "   HOMERSEKLET ADATOK FELDOLGOZASA"
40 PRINT " *****":PRINT
50 PRINT "   HOMERSEKLET ABRAZOLASA "
60 PRINT "           ES"
70 PRINT "   REGRESSZIO SZAMOLASA"
80 PRINT "AZ Y= A1+B1*X+B2*X↑0.5 EGYENLET ALAPJAN":PRINT
90 PRINT "   GEOTERMIKUS GRADIENS SZAMOLASA"
100 PAUSE 10:PRINT "□"
110 INPUT "TALPHOMERSEKLETI ADATOK SZAMA";N:PRINT
120 INPUT "KIFOLYOVIZ HOMERSEKLETI ADATOK SZAMA";M:PRINT
130 DIM A(4,N):DIM K(2,N):DIM A$(4,N)
140 DIM B(4,M):DIM L(3,M):DIM B$(4,M):DIM D(1,15)
150 FOR I=1 TO N:PRINT
160 PRINT "TALPHOMERSEKLET ADATOK:"
170 INPUT "FURAS SZAMA";A(0,I)
180 INPUT "TALPMELYSEG /M/";A(1,I)
190 INPUT "TALPHOMERSEKLET /C FOK/";A(2,I)
200 NEXT I
210 REM "TALPHOMERSEKLETI ADATOK BEOLVASASANAK VEGE"
220 FOR I=1 TO N
230 K=A(1,I)
240 SX=SX+K
250 SN=SN+K↑2
260 ZX=ZX+K↑0.5
270 ZP=ZP+K↑1.5
280 L=A(2,I)
290 SY=SY+L
300 SM=SM+L↑2
310 SS=SS+K*L
320 SR=SR+K↑0.5*L
330 NEXT I
340 QX=SN-SX↑2/N
350 QY=SM-SY↑2/N
360 SP=ZP-SX*ZX/N
370 QZ=SX-ZX↑2/N
380 PX=SS-SX*SY/N
390 PZ=SR-ZX*SY/N
400 B1=(QZ*PX-SP*PZ)/((QX*QZ)-(SP↑2))
410 B2=(QX*PZ-SP*PX)/((QX*QZ)-(SP↑2))
420 A1=(SY-B1*SX-B2*ZX)/N
430 R2=(B1*PX+B2*PZ)/QY
440 REM "REGRESSZIO SZAMITAS VEGE"
450 FOR I=1 TO N
460 A(3,I)=A1+B1*A(1,I)+B2*A(1,I)↑0.5
470 ZU=10*(A(3,I)+0.05):ZU=(INT(ZU))/10:A(3,I)=ZU
480 A(4,I)=(A(2,I)-11)/A(1,I)*1000
490 EL=10*(A(4,I)+0.05):EL=(INT(EL))/10:A(4,I)=EL

```

```

500 NEXT I
510 OPEN 4,4
520 PRINT#4,SPC(50)"1. TABLAZAT":PRINT 4,
530 PRINT#4,"FURAS"SPC(11)"MELYSEG","TALPHO" SPC(7)"SZAM.HOM."SPC(7)"VAL.GG"
540 FOR P=1 TO N
550 IF A(0,P)<10 THEN K(0,P)=2:GOTO 570
560 IF A(0,P)<100 THEN K(0,P)=1
570 IF A(1,P)<100 THEN K(1,P)=2:GOTO 590
580 IF A(1,P)<1000 THEN K(1,P)=1
590 IF A(4,P)<100 THEN K(2,P)=1
600 NEXT P
610 FOR I=1 TO 4
620 FOR P=1 TO N
630 IF A(I,P)=INT(A(I,P)) THEN GOTO 650
640 A$(I,P)=STR$(A(I,P)):GOTO 660
650 A$(I,P)=STR$(A(I,P))+".0"
660 NEXT P
670 NEXT I
680 FOR P=1 TO N
690 PRINT#4,SPC(K(0,P))A(0,P),SPC(K(1,P))A$(1,P),A$(2,P),A$(3,P),SPC(K(2,P))A$(4,P)
700 NEXT P
710 FOR I=1 TO N:V=V+A(4,I):NEXT I
720 VG=V/N
730 FOR P=1 TO M:PRINT
740 PRINT "KIFOLYOVIZ HOMERSEKLET ADATOK:"
750 INPUT "FURAS SZAMA";B(0,P)
760 INPUT "SZURO KOZEPE /M/";B(1,P)
770 INPUT "VIZHOMERSEKLET /C FOK/";B(2,P)
780 INPUT "VIZHOZAM (L/PERC)";B(3,P)
790 NEXT P
800 FOR I=1 TO M
810 K=5*(B(3,I)/1000)↑-0.71
820 H=B(2,I)+K*B(1,I)/1000
830 B(4,I)=(H-11)/B(1,I)*1000
840 ZU=(B(4,I)+0.05)*10:ZU=(INT(ZU))/10
850 EL=(B(1,I)+0.05)*10:EL=(INT(EL))/10
860 B(4,I)=ZU:B(1,I)=EL
870 NEXT I
880 PRINT#4,:PRINT#4,
890 PRINT#4,SPC(50)"2. TABLAZAT":PRINT#4,
900 PRINT#4,"FURAS"SPC(11)"MELYSEG","VIZHOM."SPC(7)"VIZHOZAM"SPC(7)"LATSZ.GG"
910 FOR P=1 TO M
920 IF B(0,P)<10 THEN L(0,P)=2:GOTO 940
930 IF B(0,P)<100 THEN L(0,P)=1
940 IF B(1,P)<100 THEN L(1,P)=2:GOTO 960
950 IF B(1,P)<1000 THEN L(1,P)=1
960 IF B(3,P)<10 THEN L(2,P)=3:GOTO 990
970 IF B(3,P)<100 THEN L(2,P)=2:GOTO 990
980 IF B(3,P)<1000 THEN L(2,P)=1
990 IF B(4,P)<100 THEN L(3,P)=1
1000 NEXT P
1010 FOR I=1 TO 4

```



```

1010 FOR I=1 TO 4
1020 FOR P=1 TO M
1030 IF B(I,P)=INT(B(I,P)) THEN GOTO 1050
1040 B$(I,P)=STR$(B(I,P)):GOTO 1060
1050 B$(I,P)=STR$(B(I,P))+".0"
1060 NEXT P
1070 NEXT I
1080 FOR P=1 TO M
1090 PRINT#4,SPC(L(0,P))B(0,P),SPC(L(1,P))B$(1,P),B$(2,P),
1100 PRINT#4,SPC(L(2,P))B$(3,P),SPC(L(3,P))B$(4,P)
1110 NEXT P
1120 FOR I=1 TO M:L=L+B(4,I):NEXT I
1130 LG=L/M
1140 PRINT#4,:PRINT#4,
1150 PRINT#4,"VALOSAGOS GEOTERMIKUS GRADIENS ATLAGA =" ;VG
1160 PRINT#4,"LATSZOLAGOS GEOTERMIKUS GRADIENS ATLAGA =" ;LG
1170 PRINT#4,:PRINT#4,
1180 PRINT#4,"AZ  $Y=A_1+B_1*X+B_2*X^2$  EGYENLET PARAMETEREI"
1190 PRINT#4, , ,SPC(3)"A1=";A1
1200 PRINT#4, , ,SPC(3)"B1=";B1
1210 PRINT#4, , ,SPC(3)"B2=";B2
1220 PRINT#4, , ,SPC(3)"R2=";R2:PRINT#4,:PRINT#4,
1230 REM "SZAMITASOK ES EREDMENEINEK KOZLESE BEFEJEZODOTT"
1240 PRINT#4,"TALPHOMERSEKLET A MELYSEG FUGGVENYEBEN":PRINT#4,
1250 FOR I=1 TO N-1
1260 FOR P=I+1 TO N
1270 IF A(1,I)<A(1,P) GOTO 1320
1280 C=A(1,I):E=A(2,I):G=A(3,I)
1290 D=A(1,P):F=A(2,P):H=A(3,P)
1300 A(1,I)=D:A(2,I)=F:A(3,I)=H
1310 A(1,P)=C:A(2,P)=E:A(3,P)=G
1320 NEXT P
1330 NEXT I
1340 FOR I=1 TO N
1350 A(1,I)=20+(A(1,I)*15)/100
1360 A(2,I)=80+(A(2,I)*18)/10
1370 A(3,I)=80+(A(3,I)*18)/10
1380 NEXT I
1390 HIRES 0,13
1400 REM "FUGG.KOORD.TG."
1410 LINE 80,0,80,200,1
1420 FOR I=20 TO 200 STEP 15
1430 LINE 75,I,80,I,1
1440 NEXT I
1450 REM "VSZ.KOORD.TG."
1460 LINE 70,20,240,20,1
1470 FOR I=80 TO 240 STEP 18
1480 LINE I,15,I,20,1
1490 NEXT I
1500 LINE 67,95,80,95,1
1510 LINE 67,170,80,170,1
1520 REM "TENGELEK MEGIRASA"

```

```

1530 TEXT 48,92,"500",1,1,8
1540 TEXT 41,167,"1000",1,1,8
1550 TEXT 48,54,"250",1,1,8
1560 TEXT 48,129.5,"750",1,1,8
1570 TEXT 55,190,"M",1,1,8
1580 TEXT 243,3,"C",1,2,0
1590 CIRCLE 240,3,3,4,1
1600 TEXT 100,5,"20",1,1,8
1610 TEXT 144,5,"40",1,1,8
1620 TEXT 180,5,"60",1,1,8
1630 TEXT 217,5,"80",1,1,8
1640 FOR I=1 TO N
1650 PLOT A(2,I),A(1,I),1
1660 NEXT I
1670 FOR P=1 TO N-1
1680 LINE A(3,P),A(1,P),A(3,P+1),A(1,P+1),1
1690 NEXT P
1700 COPY
1710 REM "ELOSZLAS VIZSGALAT"
1720 FOR I=1 TO N:K=10
1730 FOR P=1 TO 15
1740 IF A(4,I)<K THEN D(0,P)=D(0,P)+1:P=15
1750 K=K+10:NEXT P
1760 NEXT I
1770 FOR I=1 TO M:K=10
1780 FOR P=1 TO 15
1790 IF B(4,I)<K THEN D(1,P)=D(1,P)+1:P=15
1800 K=K+10:NEXT P
1810 NEXT I
1820 FOR I=0 TO 1
1830 FOR P=1 TO 15
1840 D(I,P)=180-6.5*D(I,P)
1850 NEXT P
1860 NEXT I
1870 FOR J=0 TO 1
1880 IF J=0 THEN PRINT#4, "VALOSAGOS GEOTERMIKUS GRADIENS ELOSZLASA":GOTO 1900
1890 PRINT#4, "LATSZOLAGOS GEOTERMIKUS GRADIENS ELOSZLASA"
1900 PRINT#4,
1910 HIRES 0,3
1920 LINE 20,0,20,200,1
1930 FOR I=167 TO 5 STEP -13
1940 LINE 20,I,23,I,1
1950 NEXT I
1960 LINE 0,180,320,180,1
1970 FOR I=20 TO 310 STEP 20
1980 LINE I,180,I,177,1
1990 NEXT I
2000 TEXT 283,189,"C/KM",1,1,8
2010 CIRCLE 282,186,3,4,1
2020 TEXT 1,15,"DB",1,1,8
2030 TEXT 52,183,"20",1,1,8
2040 TEXT 92,183,"40",1,1,8

```

```

2050 TEXT 132,183,"60",1,1,8
2060 TEXT 172,183,"80",1,1,8
2070 TEXT 209,183,"100",1,1,8
2080 TEXT 249,183,"120",1,1,8
2090 TEXT 5,144.5,"5",1,1,8
2100 TEXT 0,112,"10",1,1,8
2110 TEXT 0,79,"15",1,1,8
2120 TEXT 0,47,"20",1,1,8
2130 P=20
2140 FOR I=1 TO 15
2150 LINE P,180,P,D(J,I),1
2160 LINE P,D(J,I),P+20,D(J,I),1
2170 LINE P+20,180,P+20,D(J,I),1
2180 P=P+20
2190 NEXT I
2200 COPY
2210 NEXT J
2220 CLOSE 4
2230 END

```




Fig. 6. ábra: Folyamatábra. — Flow chart

A program első felében történnek az alapszámítások, a második részben az ábrázoláshoz szükséges számítások és az ábrák megrajzolatása kerül sorra.

A valóságos geotermikus gradiens számolásához szükséges adatok:

- fúrás száma /azonosítás miatt/
- talpmélység
- talphőmérséklet.

Ezekből az adatokból az $y=a+b \cdot x+c \cdot x^2$ egyenlet alapján történik a regresszió számolása és az egyes talpmélységekre a regressziós görbe alapján számolt hőmérséklet megadása. A valóságos geotermikus gradiens meghatározása a

$$GG(v) = \frac{T(t) - T(f)}{H}$$

képlet (in DÖVÉNYI et al. 1983) segítségével történik, ahol

$$\begin{aligned} T(t) &= \text{a mért talphőmérséklet} && ^\circ\text{C} \\ T(f) &= \text{felszíni középhőmérséklet} && ^\circ\text{C} \\ H(t) &= \text{talpmélység} && \text{m} \end{aligned}$$

A felszíni középhőmérséklet értéke 11 $^\circ\text{C}$ a programban. /A 400. sorban kell javítani, ha más értékkel akarunk számoltatni./

A látszólagos geotermikus gradiens számolásához szükséges adatok:

- fúrás száma /azonosítás miatt/
- szűrő küzepe
- kifolyó víz hőmérséklete
- mérési idő vízhozam

A látszólagos geotermikus gradiens számolása ugyanolyan módszerrel történik, mint a valóságos:

$$GG(l) = \frac{T(k) - T(f)}{H}$$

képlet segítségével, ahol

$$T(k) = \text{korrigált vízhőmérséklet } ^\circ\text{C}$$

$$T(k) = T(v) + h \cdot gg(h)$$

$$\begin{aligned} T(v) &= \text{mért vízhőmérséklet} && ^\circ\text{C} \\ h &= \text{vonatkoztatási mélység} && / \text{szűrő} \\ &&& \text{küzepe/} && \text{kn} \\ H &= \text{vonatkoztatási mélység} && / \text{szűrő} \\ &&& \text{küzepe/} && \text{m} \end{aligned}$$

A hűlési gradiens LIEBE P. szerint (in GÁLFI J. - STEGENA L. 1977):

$$gg(h) = 5 \cdot Q - 0,71 \quad ^\circ\text{C/km}$$

ahol

Q = vízhozam, a hőmérsékletmérés idején m^3/perc

A számítási eredmények a kiindulási adatokkal együtt táblázatos formában kerülnek kiírásra. A program a gradiensek átlagát is számolja, és ezeket is kiírja.

A regresszió számításához felhasznált egyenlet paramétereinek közlése után az első ábra a talphőmérsékleteket ábrázolja mélység függvényében ill. a pontokra illeszkedő görbét. A geotermikus gradiensek eloszlását külön-külön oszlopdiagram mutatja. /Mivel az ábrák egy meghatározott adatsor bemutatására készültek, szükséges lehet kisebb-nagyobb változtatás/.

Figyelem: Mielőtt a program begépelésére sor kerülne, a Simon's Basic segédprogram beolvastatása szükséges /ábrázolások miatt/. A 690-es sort csak akkor lehet begépelni, ha a SPC(helyett rövidítést /az S és Shift gomb lenyomásával egyidőben lenyomott P/ használunk.